
Описывается реализация мультискобки Кругликова–Лычагина в
системе компьютерной алгебры Maple. Приводится пример ис-

Введение

В работе предлагается реализация мультискобки Кругликова–
Лычагина, которая используется для решения вопроса о формаль-
ной интегрируемости системы дифференциальных уравнений [1].
Такие задачи часто возникают в дифференциальной геометрии,
теории дифференциальных уравнений и геометрической теории
управления.

Рассмотрим систему m + 1 дифференциального уравнения
в частных производных порядка s

(1) E :

{
Fi

(
x, u, . . . ,

∂|σ|u

∂xσ
, . . .

)
= 0, i = 1, . . . ,m + 1,

где x = (x1, x2, . . . , xn) — вектор размерности n; u =
(u1, u2, . . . , um) — вектор-функция от x; σ = (σ1, . . . , σn) — муль-
тииндексы длины |σ| = σ1 + . . . + σn. В этих обозначениях под

∂|σ|ui/∂xσ мы понимаем частную производную
∂|σ|ui

∂xσ1
1 ∂xσ2

2 . . . ∂xσn
n

.

Для упрощения записи введем обозначение:

pk
σ =

∂|σ|uk

∂xσ
, k = 1, . . . ,m.

Под оператором полной производной мы понимаем оператор
d

dxj
=

∂

∂xj
+
∑
k,σ

pk
σ+1j

∂

∂pk
σ

,

где σ + 1j — мультииндекс, полученный из мультииндекса σ уве-
личением его j-й компоненты на единицу.

Рассмотрим дифференциальное уравнение порядка s вида
(2) F

(
x, u, p1

σ1
, . . . , pm

σm

)
= 0.

Определение 1. Линеаризацией [2] уравнения (2) называет-
ся вектор дифференциальных операторов
(3) `(F ) =

(
L1, . . . , Lm

)
,

где

Lk =
∑

σ

∂F

∂pk
σ

d|σ|

dσx
, k = 1, . . . ,m,

d|σ|

dσx
=

dσ1

dσ1x1
◦ . . . ◦ dσn

dσnxn
,

◦ — композиция дифференциальных операторов, a
dσj

dσjxj
, j = 1, . . . , n

— оператор полной производной порядка σj по переменной xj .
Линеаризуя каждое уравнение системы (1), получим линеа-

ризованную систему в виде:

(4)


L1

1 L2
1 . . . Lm

1
...

...
. . .

...
L1

m L2
m . . . Lm

m

L1
m+1 L2

m+1 . . . Lm
m+1


 u1

...
um

 =

 0
...
0

 ,

где Lj
i — j-я компонента вектора `(Fi).

Пусть Di =
(
D1

i , . . . ,Ds
i

)
, i = 1, . . . , s, где Dj

i — линейный
дифференциальный оператор. Множество линейных дифферен-
циальных операторов образует некоммутативную алгебру отно-
сительно композиции.

Определение 2. Некоммутативным определителем
〈D1, . . . ,Dm〉 называется любое (выбранное и зафиксирован-
ное) обобщение определителя для алгебры дифференциальных
операторов:

〈D1, . . . ,Dm〉 = Ndet

 D1
1 D2

1 . . . Dm
1

...
...

. . .
...

D1
m D2

m . . . Dm
m


=
∑

α∈Sm

(−1)α D1
α(1) . . .Dm

α(m),

где Sm — множество всех перестановок из m элементов; (−1)α

— четность перестановки α ∈ Sm.
Впоследствии, вычисляя некоммутативный определитель,

мы всегда будем раскладывать его по первому столбцу.
Определение 3. Мультискобкой Кругликова–Лычагина

функций F1, F2, . . . , Fm+1 называется функция на пространстве
джетов [2]

(5) {F1, F2, . . . , Fm+1}
def
=

m+1∑
i=1

〈
L1, . . . , L̂i, . . . , Lm+1

〉
Fi,

где Li = `(Fi) =
(
L1

i , L
2
i , . . . , L

m
i

)
, i = 1, . . . ,m+1, а L̂i означает

пропуск i-й компоненты.
Основной результат, который мы используем для исследова-

ния формальной интегрируемости системы дифференциальных
уравнений (1), состоит в следующем:

1) Если система E формально интегрируема, то мультискобка

{F1, . . . , Fm, Fm+1}

равна нулю в силу системы E .

2) Если система E находится в полном пересечении ([1, 3]),
то E формально интегрируема тогда и только тогда, когда
мультискобка

{F1, . . . , Fm, Fm+1}
равна нулю в силу системы E .

Если рассматриваемая система состоит из более чем m + 1 урав-
нений, то необходимо рассмотреть скобки для всевозможных на-
боров из m+1 уравнений. Подробное изложение этих результатов
можно найти в [5].

1. Вспомогательные процедуры

Для реализации вычисления формулы (5) нам необходимы
следуюшие процедуры:

1) операции над дифференциальными операторами: сложение,
умножение на функцию и композиция;

2) вычисление некоммутативного определителя матрицы;

3) вычисление линеаризации системы дифференциальных
уравнений.

1.1. Операции над дифференциальными операторами
Для представления дифференциальных операторов в про-

грамме мы выбрали определенный в Maple тип table [6], кото-
рый представляет собой ассоциативный массив. Ключом в нашем
массиве будет мультииндекс σ, а значением — алгебраическое вы-
ражение, стоящее перед соответствующей индексу частной про-
изводной.

Таким образом, каждая пара ключ-значение
((σ1, . . . , σn) , Fσ (x1, . . . , xn)) представляет оператор
Fσ (x1, . . . , xn) ∂|σ|/∂xσ. Тогда весь ассоциативный массив
T представляет дифференциальный оператор∑

σ∈K(T )

Fσ (x1, . . . , xn)
∂|σ|

∂xσ
,

где K (T ) — множество ключей массива. Далее в зависимости
от контекста мы будем понимать под T и сам оператор, и ассоци-
ативный массив, которым оператор представляется.

Сложение дифференциальных операторов. В выбранном
представлении операторов сумма двух операторов

T1 =
∑

σ∈K(T1)

Fσ
∂|σ|

∂xσ
и T2 =

∑
τ∈K(T2)

Gτ
∂|τ |

∂xτ

задается следующим образом:

T1 + T2 =
∑

σ∈K(T1)\K(T2)

Fσ
∂|σ|

∂xσ
+

+
∑

τ∈K(T2)\K(T1)

Gτ
∂|τ |

∂xτ
+

∑
σ∈K(T1)∩K(T2)

(Fσ + Gσ)
∂|σ|

∂xσ
.

Формула для разности аналогична. Реализация сложения двух
операторов на языке Maple представлена ниже.

diffopPlus := proc( diffop1::table, diffop2::table )::table;
local result::table, sigma::list;
result := copy( diffop1 );
for sigma in indices( diffop2 ) do
if sigma in {indices( result )} then
result[op( sigma )] := result[op( sigma )] + diffop2[op( sigma )];

else
result[op( sigma )] := diffop2[op( sigma )];

end if;
end do;
return copy( result );

end proc;

Умножение дифференциального оператора на функцию.
Умножение оператора

T =
∑

σ∈K(T )

Fσ
∂|σ|

∂xσ

на функцию g(x1, . . . , xn) слева задается формулой∑
σ∈K(T )

gFσ
∂|σ|

∂xσ
.

Код, ее реализущий:

diffopMult := proc( diffOp::table, g )::table;
local result::table, sigma::list;
result := copy( diffOp );
for sigma in indices( diffOp ) do
result[op( sigma )] := result[op( sigma )] * g;

end do;
return copy( result ):

end proc:

Композиция дифференцирования и дифференциального мо-
нома. Пусть дан моном

M = f
∂|τ |

∂xτ
,

где τ — мультииндекс. Рассмотрим композицию дифференциро-
вания по переменной xi и монома M . Формула для вычисления
композиции вытекает из правила Лейбница и выглядит так:

∂

∂xi
◦M =

∂f

∂xi

∂|τ |

∂xτ
+ f

∂|τ |+1

∂xτ1
1 . . . ∂xτi+1

2 . . . ∂xτn
n

Процедура на Maple, вычисляющая данную композицию:

diffopDMonomial := proc( diffMon::table, varIndex::integer )::table;
local result::table, sigma::list, othersigma::list;
if nops( [indices( diffMon )] ) <> 1 then
error "Incorrect monomial has size %1, indices=%2",
nops( [indices( diffMon )] ), [indices( diffMon )]

end if;
if ivarsCount < varIndex or varIndex < 1 then
error "Incorrect variable’s index %1", varIndex;

end if;
sigma := indices( diffMon )[1];
othersigma := sigma;
othersigma[varIndex] := othersigma[varIndex] + 1;
result := copy( diffMon );
result[op( sigma )] := TDiff( diffMon[op( sigma )], varIndex );
result[op( othersigma )] := diffMon[op( sigma )];
return copy( result );

end proc:

Композиция дифференцирования и дифференциального опе-
ратора. Используя функцию diffopDMonomial, реализованную
выше, мы можем построить композицию оператора взятия част-
ной производной и дифференциального оператора. Пусть задан
оператор

T =
∑

σ∈K(T )

Mσ,

где Mσ = Fσ
∂|σ|

∂xσ . Формула для композиции записывается следу-
ющим способом:

∂

∂xi
◦ T =

∑
σ∈K(T )

∂

∂xi
◦Mσ.

Исходный код этой процедуры на Maple:

diffopDOperator := proc( diffOp::table, varIndex::integer )::table;
local result::table, sigma::list, diffMon::table;
if ivarsCount < varIndex or varIndex < 1 then
error "Incorrect variable’s index %1", varIndex;

end if;
result := table();
for sigma in indices( diffOp ) do
diffMon := table( [op( sigma ) = diffOp[op( sigma )]] );
result := diffopPlus( result, diffopDMonomial( diffMon, varIndex ) );

end do;
return result;

end proc:

Композиция дифференциального монома и дифференциаль-
ного оператора. Пусть

N = f
∂|τ |

∂xτ

— дифференциальный моном, а

T =
∑

σ∈K(T )

Mσ

— дифференциальный оператор. Композиция N ◦T строится сде-
дующим образом:

N ◦ T = N ◦
∑

σ∈K(T )

Mσ =
∑

σ∈K(T )

N ◦Mσ = f
∑

σ∈K(T )

∂|τ |

∂xτ
◦Mσ.

Реализация на Maple:

diffopComposeWithMonomial := proc( diffMon::table, diffOp::table )::table;
local result::table, diffs::list, f, diffIndex::integer, varIndex::integer;
if nops( [indices(diffMon)] ) <> 1 then
error "Incorrect monomial has size %1, indices=%2",
nops( [indices(diffMon)] ), [indices(diffMon)]

end if;
result := copy( diffOp );
diffs := indices( diffMon )[1];
f := diffMon[op( diffs )];
for varIndex from 1 to ivarsCount do
for diffIndex from 1 to diffs[varIndex] do
result := diffopDOperator( result, varIndex )

end do
end do;
result := diffopMult( result, f );
return copy( result );

end proc:

Композиция дифференциальных операторов. Имея в распо-
ряжении композицию монома и оператора, мы можем легко по-
строить композицию двух операторов. Пусть даны операторы

T1 =
∑

σ∈K(T1)

Mσ

и T2. Композицию T1 ◦ T2 мы запишем в виде:

T1 ◦ T2 =

 ∑
σ∈K(T1)

Mσ

 ◦ T2 =
∑

σ∈K(T1)

(Mσ ◦ T2) .

Процедура, реализующая композицию двух операторов, пред-
ставлена ниже.

diffopCompose := proc( diffOp1::table, diffOp2::table )::table;
local result::table, sigma::list, diffMon::table;
result := table();
for sigma in indices( diffOp1 ) do
diffMon := table( [op( sigma ) = diffOp1[ op(sigma) ]] );
result := diffopPlus( result,
diffopComposeWithMonomial( diffMon, diffOp2 ) );

end do;
diffopRemoveZeroes( result );
return copy( result );

end proc:

Применение дифференциального оператора к уравнению.
При реализации мультискобки нам необходимо применять опе-
раторы к уравнениям из исходной системы (1). Поэтому мы реа-
лизуем следующую процедуру.

Определим действие дифференциального оператора

T =
∑

σ∈K(T )

Fσ
∂|σ|

∂xσ

на функцию G(x1, x2, . . . , xn) следующей формулой:

T (G) =
∑

σ∈K(T )

Fσ
∂|σ|G

∂xσ
.

Исходный код процедуры на Maple:

diffopOnEquation := proc( diffOp::table, eq )
local result, sigma::list, F, varIndex::integer, i::integer;
result := 0;
for sigma in indices( diffOp ) do
F := eq;
result := result + diffOp[op( sigma )] * TotDiff( F,
sigma2vars( sigma ) );

end do;
return result;

end proc;

1.2. Реализация некоммутативного определителя
Имея в распоряжении построенные вспомогательные проце-

дуры, мы можем реализовать процедуру вычисления некоммута-
тивного определителя в соответствии с определением 2. Мы бу-
дем вычислять определитель, раскладывая его по первому столб-
цу. Ниже мы приводим текст процедуры Ndet на языке Maplе,
вычисляющей определитель матрицы, используя рекурсивный ал-
горитм.
Ndet := proc( A::Matrix )::table;
local N::integer, temp::Matrix, result::table, row::integer,
i::integer, j::integer, tempRow::integer, sign::boolean;
N := op( 1, A )[1];
if N <> op( 1, A )[2] then
error "Matrix in Ndet is not square, %1 <> %2", N, op( 1, A )[2];

end if;
if N = 2 then
return copy( diffopMinus( diffopCompose( A[1, 1], A[2, 2] ),
diffopCompose( A[2, 1], A[1, 2] ) ) );

end if;
temp := Matrix( N - 1, N - 1 );
result := table();
sign := true;
for row from 1 to N do
sign := not sign;
tempRow := 0;
for i from 1 to N do
if i <> row then
tempRow := tempRow + 1;
for j from 2 to N do
temp[tempRow, j - 1] := A[i, j];

end do;
end if;

end do;
if sign then
result := diffopMinus( result, diffopCompose( A[row, 1],
Ndet( temp ) ) );

else
result := diffopPlus( result, diffopCompose( A[row, 1],
Ndet( temp ) ) );

end if;
end do;
return copy( result );

end proc:

Входной параметр функции Ndet имеет тип Matrix, который
определен в Maple. Этот тип позволяет хранить матрицу, элемен-
ты которой могут иметь любой тип. В качестве результата работы
функция возвращает дифференциальный оператор, хранящийся в
ассоциативном массиве типа table.

1.3. Линеаризация
Важным этапом алгоритма вычисления мультискобки Кру-

гликова–Лычагина является вычисление линеаризации системы

уравнений (1). Чтобы линеаризовать систему дифференциальных
уравнений, необходимо линеаризовать каждое из уравнений этой
системы. Используя явную формулу из определения (1), реализу-
ем на языке Maple линеаризацию одного уравненения.

linearizeEquation := proc( F )::Array;
local result::Array, i::integer, sigmas::set, sigma;
result := Array( 1 .. dvarsCount );
for i from 1 to dvarsCount do
result[i] := table();
sigmas := allSigmas( F, dvars[i] );
for sigma in sigmas do
result[i][op( sigma )] := diff( F, dvars[i][op( sigma )] );

end do;
end do;
return result;

end proc:

Аргумент функции linearizeEquation — это выражение F, со-
держащее независимые переменные (хранятся в глобальном спис-
ке ivars) и координаты джетов, записанные через зависимые пе-
ременные (хранятся в глобальном списке dvars) с мультииндек-
сом. Эта функция возвращает объект типа Array (определенный в
Maple тип массива), содержащий дифференциальные операторы.

Размер возвращаемого массива в точности равен числу m —
количеству неизвестных функций ui, которое равно переменной
dvarsCount — длине списка dvars. На i-й позиции в этом массиве
стоит дифференциальный оператор при i-й функции в форму-
ле (3).

Встречающася в приведенном тексте функция allSigmas —
одна из вспомогательных функций, на реализации которых мы
не останавливаемся. Данная функция, анализируя выражение, пе-
редаваемое ей в первом аргументе F, ищет все записанные джета-
ми производные функции, имя которой было передано ей вторым
аргументом dvars[i], и возвращает список всех мультииндексов,
стоящих при этом имени в выражении F.

Построенную линеаризацию уравнения мы применим для
процедуры линеаризация системы уравнений вида (1):

linearizeSystem := proc( sys::list )::Matrix;
local result::Matrix, eq::Array, i::integer, j::integer, eqsCount::integer;
eqsCount := nops( sys );
result := Matrix( eqsCount, dvarsCount );
for i from 1 to eqsCount do
eq := linearizeEquation( sys[i] );

for j from 1 to dvarsCount do
result[i, j] := copy( eq[j] );

end do;
end do;
return result;

end proc;

Процедура linearizeSystem линеаризует систему уравнений,
переданную ей в списке sys. Находя линеаризацию каждого урав-
нения в этом списке, она строит матрицу дифференциальных опе-
раторов, имеющую dvarsCount столбцов и на единицу большее,
чем число уравнений, количество строк.

2. Реализация мультискобки Кругликова–Лычагина

Реализовав все вспомогательные процедуры, мы переходим
к реализации определения (3). Ниже представлен исходный текст
процедуры MultiBracket.

MultiBracket := proc( eqSys::list )
local linearization::Matrix, Ndets::Array, result, i::integer;
linearization := linearizeSystem( eqSys );
Ndets := allNdets( linearization );
result := 0;
for i from 1 to dvarsCount + 1 do
result := result + (-1)^( (i + 1) mod 2 ) *
diffopOnEquation( Ndets[i], eqSys[i] );

end do;
return simplify( result );

end proc;

Входной параметр eqSys — список уравнений исходной си-
стемы. Процедура возвращает выражение — вычисленную муль-
тискобку. Процедура allNdets принимает на вход матрицу диф-
ференциальных операторов размера m × (m + 1), а возвращает
спиcок из m некоммутативных определитей, полученных из вход-
ной матрицы.

Замечание 1. Оценим сложность вычисления нескольких
реализованных функций. Функция Ndet вычисляет определитель
матрицы разложением по столбцам, следовательно, выполняет
m! операций композиции, где m — число неизвестных функций.
Функция MultiBracket вызывает m + 1 раз функцию Ndet, а затем
применяет результат каждого вычисления m + 1 раз уравнени-
ям исходной системы, т.е. выполняет (m + 1)2 m! композиций.

137



Управление большими системами. Выпуск 33

Несмотря на то, что формула вычисления мультискобки выпол-
няет значительное количество операций, этот метод получения
условий формальной интегрируемости является довольно быст-
рым, потому что задает явную формулу. Сравнение метода скобок
с другим методом – дифференциальными базисами Гребнера –
приведено в [4].

3. Пример

Рассмотрим следующую задачу: описать все векторные по-
ля без особых точек на плоскости, первые интегралы которых
являются гармоническими функциями.

Пусть векторные поля имеют вид:

X = a(x, y)
∂

∂x
+ b(x, y)

∂

∂y
.

Задача сводится к исследованию совместности следующей систе-
мы дифференциальных уравнения в частных производных:{

∆u = 0,

X(u) = 0,

где ∆ = ∂xx +∂yy — оператор Лапласа. Согласно условию Коши–
Римана эта система эквивалентна следующей системе дифферен-
циальных уравнений первого порядка:

(6)


ux − vy = 0,

uy + vx = 0,

a(x, y)ux + b(x, y)uy = 0.
В силу того, что векторное поле X не имеет особых точек, можно
считать, что функция a(x, y) нигде не обращается в ноль. Тогда
без ограничения общности можно положить a(x, y) ≡ 1. Система
(6) примет вид:

(7)


ux − vy = 0,

uy + vx = 0,

ux + b(x, y)uy = 0.

Программа на Maple, вычисляющая скобку для системы (7):

# подключаем модуль Brackets
with( Brackets ):
# настраиваем внутренние параметры этого модуля:
# списки независимых и зависимых переменных
setup( [x, y], [u, v] ):
# задаем систему уравнений
S := [ u[1, 0] - v[0, 1], u[0, 1] + v[1, 0],
u[1, 0] + b(x, y) * u[0, 1] ]:

# вычисляем скобку и группируем члены у проивзодных u_x, u_y, u_xy
MlBr := collect( MultiBracket( S ), [ u[0, 1], u[1, 1], u[0, 2] ] );

В результате компьютерных вычислений получим мультис-
кобку Кругликова–Лычагина:
(8) M = ∆buy + 2byuyy + 2bxuxy,

где ∆ = ∂xx + ∂yy — оператор Лапласа. Ограничим скобку M на
систему (7), используя третье уравнение ux = −b(x, y)uy. Диффе-
ренцируя его по y, получим uxy = −byuy − buyy. Ограниченную
скобку обозначим

ME = (∆b− 2bxby)uy + 2(by − bbx)uyy.

Чтобы система (7) была формально интегрируема, необходимо и
достаточно, чтобы ME ≡ 0. Видно, что это условие выполнено
тогда и только тогда, когда коэффициенты при uy и uyy равны
нулю. Это условие приводит к следующей системе уравнений:

(9)

{
∆b = 2bybx,

by = bbx.

Решая ее, получим
b(x, y) = h1(y)x + h2(y),

где h1, h2 — произвольные гладкие функции.
Чтобы исследовать совместность системы (7), перепишем ее

в форме, не содержащей функцию v(x, y), и добавим к ней урав-
нение ME = 0:

(10)


uxx + uyy = 0,

ux + buy = 0,

(∆b− 2bxby)uy + 2(by − bbx)uyy = 0.

Продифференцируем второе уравнение этой системы по x и по y
и добавим полученные соотношения к системе (10):

uxx + uyy = 0,

ux + buy = 0,

uxx + bxuy + buxy = 0,

uxy + byuy + buyy = 0,

(∆b− 2bxby)uy + 2(by − bbx)uyy = 0.
Эта система эквивалентна следующей системе:

(11)



uxx + uyy = 0,

ux + buy = 0,

uxy + byuy + buyy = 0,

−(1 + b2)uyy + (bx − bby)uy = 0,

2(by − bbx)uyy + (∆b− 2bxby)uy = 0.
Необходимо рассмотреть два случая.

В первом случае, когда uy 6= 0 и uyy 6= 0, определитель
системы, составленной из четвертого и пятого уравнений систе-
мы (11), должен быть равен нулю. Это условие дает следующее
соотношение на функцию b(x, y):
(12) (b2 + 1)∆b− 2b(b2

x + b2
y) = 0.

C учетом соотношения (12) выразим из системы (11) производные
uxx и uxy через uy:

uxx = −(bx − bby) uy

b2 + 1
,

uxy = −
(

by +
b (bx − bby)

b2 + 1

)
uy.

Используя эти соотношения, условие (12) и систему (11), мож-
но показать, что условия формальной интегрируемости системы
(11):

uyyx = uxyy,

uxxy = uxyx
выполняются тождественно.

Рассмотрим второй случай. Пусть uy = uyy = 0. Тогда си-
стема (11) сводится к системе:{

uy = 0,

ux = 0,

решение которой u(x, y) = const. Таким образом мы приходим к
следующей теореме.

Теорема 1. Векторное поле

X =
∂

∂x
+ b(x, y)

∂

∂y
обладает не равным константе гармоническим первым интегра-
лом тогда и только тогда, когда функция b(x, y) удовлетворяет
следующему дифференциальному уравнению:

(b2 + 1)∆b− 2b(b2
x + b2

y) = 0.

